Eddy current damping in OSEM bodies, clamps and penultimate masses of Advanced LIGO ITM/ETM suspensions

K.A. Strain   LIGO – T050102 – 00 – K

Purpose: mainly to enable decision choice of material for ITM/ETM penultimate reaction mass OSEMS. Also informs that there are no problems due to eddy current damping in nearby metalwork.

Requirements are from T010007-03, assumptions about magnets from G010086.

Damping from a hollow cylindrical conductor, for axial motion of a small magnet
Here are a few of the fixed parameters and the necessary equations

```
r =.; z =.; (*coordinates*)
 s =.; (*conductivity*)
 μ = 4 Pi 10^-7; (*vacuum permeability*)
 pm = 0.007;(*approx dipole moment of magnet from T000119-00 assume same magnets used*)
 α = 3 μ pm / (4 Pi);   (*hide some notation - α comes from standard works*)
 Br = α r z / (r^2 + z^2)^(5/2);(*radial dipole field - for reference, not actually used here*)
 ev = 2 Pi r Br; (*emf per unit velocity*)
 (*damping is power/velocity^2, power is emf^2/resistance*)
 dR = 2 Pi r / (s); (*resistace for unit area at radius r*)
 dB = ev^2 / dR; (*damping integrand is ev^2/resistance - integrate up dr dz to get b - damping*)
```

Now the case of a steel cylinder

```
s = 1.4 10^6; (*conductivity of stainless - roughly correct for all the types we use*)

r1 = 0.001; (*inner radius*)
r2 = 0.10; (*outer radius - normally not very imporant,
 set a large value for an upper limit to the damping*)
z1 = 0.001; (*distance to nearest face*)
z2 = 0.05; (*distance to far face - normally not very imporant,
 set a large value for an upper limit to the damping*)
b = Integrate[dB, {z, z1, z2}, {r, r1, r2}];

ϕ = 20 Pi b / (40 4 Pi^2)  (*for ωo = 2 Pi 1 Hz resonance, m= 40 kg, and find phi at ω = 2 Pi 10 Hz*)

0.0000153828 + 0. i
```

Some pre-calculated values to give a guide to damping as a function of radius

```
ANS = List[0.0000966526885960794, 4.639764858828375*^-6 , 1.8465655731017687*^-7, 5.009031494531008*^-9]
 ListPlot[Log[10, ANS], PlotJoined → True]
(*upper limits for 1, 3.2, 10 and 32 mm r1, with 1 mm z1 and large r2,z2  - non evaluating cell*)
```

$\{0.0000966527,\ 4.63976\times10^{-6},\ 1.84657\times10^{-7},\ 5.00903\times10^{-9}\}$



- Graphics -

Discussion

There is a factor of at least 200 isolation from penultimate to test masses, the viscous damping may be 5x10^4 larger at the penultimate mass than would be permitted at the test mass (see note 2 below). A guide figure for viscous loss (see note 1 below) is 1e-11 at the TM and so 5e-7 at the PM. There are 4 magnets, so the loss per magnet should not exceed about 10^-7. Looking at the graph that corresponds to close to the 3rd point in the list - corresponding to 10mm radius. Any steel rings larger than about 10mm radius will, therefore, not be problematic with our standard magnets. The same would apply for any radius of ring in any plane more than about 1cm from the magnet. The implication is that the penultimate reaction mass itself is safe (the holes being much larger than this). The OSEM clamps are much more than 10 mm away and any case slotted giving a substantial margin (not calculated). The OSEM body is also slotted, but is much closer to the magnet. An FE calculation would be required to establish whether a steel OSEM body could be employed (Aluminium, with 40 times higher damping, is almost certainly not permitted even with a slot). The safest alternative is, however, to employ an alumina part. Conclusion: use an insulating (alumina) OSEM body at the penultimate stage of the ETM/ITM quad suspensions. All other components at all stages are acceptable as designed.

Note 1: Thermal noise from viscous damping

```
kb = 1.38 10^-23; T = 300; ω = 2 Pi 10;  m = 40; g = 9.81; l = 0.6;  k = m g / l; ω0 = Sqrt[g / l];
 (*to allow rough estimate of damping as a function of viscous loss*)
 γ = k ϕ / ω; (*viscous damping*)
 NT2 = 4 kb T γ / (m^2 ((ω^2 - ω0^2)^2 + γ ω^2)); NT = Sqrt[NT2];
```

```
Plot[Log[10, NT], {ϕ, 10^-11, 5 10^-11}]
```



- Graphics -

Note 2: Estimate of required Q at PM (calculate isolation from PM to TM and square it)

```
(*isolation ratio from MATLAB QUAD ITM/ETM model is 47 dB *)
isol = 10^(47/20.)
 qratio = 10^(47/10.)
```

223.872

50118.7